```
/* Project 2B - ADAE Creation */;
options validvarname=upcase nofmterr;

%let SDTMPATH = /home/u64124548/sasuser.v94/ADAM - Creation/SDTM_Data_Input;
%let OUTPATH  = /home/u64124548/sasuser.v94/ADAM - Creation/ADAM_Output_LogFiles;
%let VALPATH  = /home/u64124548/sasuser.v94/ADAM - Creation/Validation_Data_ADAM;

libname adam "&OUTPATH";

proc printto log="&OUTPATH/ADAE.log" new;
run;

libname xae     xport "&SDTMPATH/AE.xpt";
libname xsuppae xport "&SDTMPATH/SUPPAE.xpt";

proc copy in=xae out=work; run;
proc copy in=xsuppae out=work; run;

proc sort data=adam.adsl out=adsl;
    by USUBJID;
run;

proc sort data=suppae out=suppae_sort;
    by STUDYID RDOMAIN USUBJID IDVAR IDVARVAL;
run;

proc transpose data=suppae_sort out=suppae_t(drop=_NAME_ _LABEL_);
    by STUDYID RDOMAIN USUBJID IDVAR IDVARVAL;
    id QNAM;
    var QVAL;
run;

data suppae_t;
    length AERELOTH AETRTOTH $200 AEDTHDTC $20 DTHAUTYN $2 AEHDTC $20
           AEHLGT AEHLT AELLT $200 AELLTCD $8;
    set suppae_t;

    AESEQ=input(IDVARVAL,best.);

    keep USUBJID AESEQ AEHLGT AEHLT AELLT AELLTCD
         AERELNS1 AERELNS2 AERELOTH
         AEACNOTH AEACNOT1 AEACNOT2 AEACNOT3 AETRTOTH
         AEDTHDTC DTHAUTYN AEHDTC;
run;

proc sort data=ae;       by USUBJID AESEQ; run;
proc sort data=suppae_t; by USUBJID AESEQ; run;

data ae_all;
    merge ae(in=a) suppae_t;
    by USUBJID AESEQ;
    if a;
run;

proc sort data=ae_all; by USUBJID; run;


data adam.adae(label="Adverse Event");
```

```
    length
        STUDYID $8 USUBJID $21 SUBJID $10 SITEID $20 INVNAM $50 INVID $5
        AGEU $5 AGEGR1 $8 SEX $1 RACE $32 ETHNIC $22 COUNTRY $3
        TRT01P $10 TRT01A $10 REMISS $40
        ITTFL $1 SAFFL $1 EFFL $1 CA125FL $1 TRTDCFL $1 COMPLFL $1
        STDDCFL $1 SFUDCFL $1 SFUFL $1
        SRCDOM $2
        AETERM $200 AEMODIFY $200 AEDECOD $200 AEBODSYS $200
        AEHLGT $200 AEHLT $200 AELLT $200 AELLTCD $8
        TRTSDTC $20 TRTEDTC $20
        AESTDTC $20 AESTRTPT $10 AEENDTC $20 AEENRTPT $10
        AESER $2 AEREL $1 AERELOTH $200
        AETOXGR $1 AEACN $20 AETRTOTH $200
        AESDTH $2 AEDTHDTC $20 DTHAUTYN $2 AESLIFE $2 AESHOSP $3 AEHDTC $20
        AESDISAB $20 AESCONG $20 AESMIE $20 TRTEM $1;

    merge ae_all(in=a)
          adsl(in=b keep=STUDYID USUBJID SUBJID SITEID INVNAM INVID AGE AGEU AGEGR1 AGEGR1N
                        SEX RACE ETHNIC COUNTRY RANDDT TRT01P TRT01PN TRT01A TRT01AN
                        REMISS REMISSN ITTFL SAFFL EFFL CA125FL TRTDCFL COMPLFL
                        STDDCFL SFUDCFL SFUFL TRTSDT TRTSDTC TRTEDT TRTEDTC);
    by USUBJID;

    if a;

    SRCDOM = DOMAIN;
    SRCSEQ = AESEQ;

    if length(strip(substr(AESTDTC,1,10)))=10 then AESDT=input(substr(AESTDTC,1,10), yymmdd10.);
    else AESDT=.;

    if length(strip(substr(AEENDTC,1,10)))=10 then AEEDT=input(substr(AEENDTC,1,10), yymmdd10.);
    else AEEDT=.;

    if AESDT ne . and TRTSDT ne . then do;
        if AESDT >= TRTSDT then AESDY = AESDT - TRTSDT + 1;
        else AESDY = AESDT - TRTSDT;
    end;

    AETOXGR = strip(AETOXGR);
    AETOXGRN = input(AETOXGR,best.);

    if upcase(strip(AERELNST)) ne 'MULTIPLE' then AERELOTH = AERELNST;
    else AERELOTH = catx('; ', AERELNS1, AERELNS2);

    if strip(AEACNOTH) ne '' and upcase(strip(AEACNOTH)) ne 'MULTIPLE' then do;
        if AECONTRT='Y' and upcase(strip(AEACNOTH)) not in ('NONE','') then
            AETRTOTH = catx('; ', AEACNOTH, 'MEDICATION');
        else if AECONTRT='Y' then AETRTOTH = 'MEDICATION';
        else AETRTOTH = AEACNOTH;
    end;
    else if upcase(strip(AEACNOTH))='MULTIPLE' then
        AETRTOTH = catx('; ', AEACNOT1, AEACNOT2, AEACNOT3);
    else if AECONTRT='Y' then AETRTOTH = 'MEDICATION';

    TRTEM='Y';

    if TRTSDT ne . then do;
        if AESDT ne . then do;
            if AESDT < TRTSDT then TRTEM='N';
```

```
            else TRTEM='Y';
        end;
        else do;
            ae_year  = input(substr(AESTDTC,1,4), best.);
            ae_month = input(substr(AESTDTC,6,2), best.);
            trt_year = year(TRTSDT);
            trt_month= month(TRTSDT);

            if ae_year ne . and ae_month ne . then do;
                if ae_year < trt_year then TRTEM='N';
                else if ae_year = trt_year and ae_month < trt_month then TRTEM='N';
                else TRTEM='Y';
            end;
            else if ae_year ne . then do;
                if ae_year < trt_year then TRTEM='N';
                else TRTEM='Y';
            end;
            else TRTEM='Y';
        end;
    end;

    drop ae_year ae_month trt_year trt_month;

    format RANDDT TRTSDT TRTEDT AESDT AEEDT is8601da.;

    label
        STUDYID   = "Study ID"
        USUBJID   = "Unique Subject Identifier"
        SUBJID    = "Subject Identifier for the Study"
        SITEID    = "Study Site Identifier"
        INVNAM    = "Investigator Name"
        INVID     = "Investigator Identifier"
        AGE       = "Age"
        AGEU      = "Age Units"
        AGEGR1    = "Age Group 1 (Char)"
        AGEGR1N   = "Age Group 1 (Num)"
        SEX       = "Sex"
        RACE      = "Race"
        ETHNIC    = "Ethnicity"
        COUNTRY   = "Country"
        RANDDT    = "Randomization Date"
        TRT01P    = "Planned Treatment"
        TRT01PN   = "Planned Treatment Number"
        TRT01A    = "Actual Treatment"
        TRT01AN   = "Actual Treatment Number"
        REMISS    = "Current Remission Status (char)"
        REMISSN   = "Current Remission Status (Num)"
        ITTFL     = "Intent-to-Treat Population Flag"
        SAFFL     = "Safety Population Flag"
        EFFL      = "Efficacy-Evaluable Population Flag"
        CA125FL   = "Efficacy-Evaluable CA125 Population Flag"
        TRTDCFL   = "Treatment Discontinuation Flag"
        COMPLFL   = "Study Period Completers Flag"
        STDDCFL   = "Study Period Discontiuation Flag"
        SFUDCFL   = "Follow-Up Period Discontinuation Flag"
        SFUFL     = "Survival Follow-Up Period Entry Flag"
        SRCDOM    = "Source Domain"
        SRCSEQ    = "Source Sequence Number"
        AETERM    = "Reported Term for the Adverse Event"
        AEMODIFY = "Modified Reported Term"
```

```
        AEDECOD  = "Dictionary-Derived Term"
        AEBODSYS = "Body System or Organ Class"
        AEHLGT   = "High Level Group Term"
        AEHLT    = "High Level Term"
        AELLT    = "Low Level Term"
        AELLTCD  = "Low Level Term Code"
        TRTSDT   = "First Treatment (GDC) Date (num)"
        TRTSDTC  = "First Treatment (GDC) Date (char)"
        TRTEDT   = "Last Treatment (GDC) Date (num)"
        TRTEDTC  = "Last Treatment (GDC) Date (char)"
        AESTDTC  = "Start Date/Time of Adverse Event"
        AESTRTPT = "Start Relative to Reference Time Point"
        AEENDTC  = "End Date/Time of Adverse Event"
        AEENRTPT = "End Relative to Reference Time Point"
        AESDT    = "AE Start Date"
        AEEDT    = "AE End Date"
        AESDY    = "Relative Start Day of AE"
        AESER    = "Serious Event"
        AEREL    = "Causality"
        AERELOTH = "Relationship to Non-Study Treatment"
        AETOXGR  = "Standard Toxicity Grade"
        AETOXGRN = "Standard Toxicity Grade (num)"
        AEACN    = "Action Taken with Study Treatment"
        AETRTOTH = "Treatment for AE"
        AESDTH   = "Results in Death"
        AEDTHDTC = "Death Date"
        DTHAUTYN = "Was Autopsy Performed"
        AESLIFE  = "Is Life Threatening"
        AESHOSP  = "Requires or Prolongs Hospitalization"
        AEHDTC   = "Hospitalization Admission Date"
        AESDISAB = "Persist or Signif Disability/Incapacity"
        AESCONG  = "Congenital Anomaly or Birth Defect"
        AESMIE   = "Other Medically Important Serious Event"
        TRTEM    = "Treatment Emergent";

    keep STUDYID USUBJID SUBJID SITEID INVNAM INVID AGE AGEU AGEGR1 AGEGR1N
         SEX RACE ETHNIC COUNTRY RANDDT TRT01P TRT01PN TRT01A TRT01AN
         REMISS REMISSN ITTFL SAFFL EFFL CA125FL TRTDCFL COMPLFL STDDCFL
         SFUDCFL SFUFL SRCDOM SRCSEQ AETERM AEMODIFY AEDECOD AEBODSYS
         AEHLGT AEHLT AELLT AELLTCD TRTSDT TRTSDTC TRTEDT TRTEDTC
         AESTDTC AESTRTPT AEENDTC AEENRTPT AESDT AEEDT AESDY AESER AEREL
         AERELOTH AETOXGR AETOXGRN AEACN AETRTOTH AESDTH AEDTHDTC DTHAUTYN
         AESLIFE AESHOSP AEHDTC AESDISAB AESCONG AESMIE TRTEM;
run;

proc sort data=adam.adae;
    by USUBJID SRCSEQ;
run;


libname xadae xport "&OUTPATH/ADAE.xpt";

proc copy in=adam out=xadae;
    select adae;
run;


libname vadae xport "&VALPATH/ADAE.xpt";

proc sort data=vadae.adae out=base_adae;
```

```
    by USUBJID SRCSEQ;
run;

proc sort data=adam.adae out=comp_adae;
    by USUBJID SRCSEQ;
run;

ods rtf file="&OUTPATH/ADAE_PROC_COMPARE.rtf";

title "PROC COMPARE: Validation ADAE vs Created ADAE";
```